
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 11.2.1a which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_11_2_1a
Test Group:
/11/UTRAN34_NISPC.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D15wk10 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
MediaTek K2 Handset
Verification Status:
PASS
4 Corrections required for test case 11.2.1a
4.1 Introduction

This section describes the changes required to make test case 11.2.1a run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D15wk10 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 fl_TC_11_2_1a_Body()
	Record Name
	fl_TC_11_2_1a_Body()

	Reason for change
	In the TTCN implementation of the  function fl_TC_11_2_1a_Body() the variable v_TFT2 is being incorrectly updated before it is sent in the Modify PDP Context REQ according to Step 19-20 of the expected test sequence in the Prose. Basically, the position of TOS_TC is being set incorrectly. This needs to be corrected.

Later, in the TTCN implementation for Steps 27-28 where the SS sends a Modify PDP Context REQ to remove the packet filter, the contents of the  variable v_TFT1 are being set incorrectly. This needs to be corrected. Also, the Modify PDP Context REQ sent at 19 should contain the PCO which sets the BCM to MS/NW. If this is not done, then the UE will assume the BCM to be MS only, and this will create issue when at Step 27 another Modify PDP Context REQ is sent to remove a UL TFT filter.
Also in the current TTCN implementation, some commented code is lying around which has been removed (marked by green in Before change screenshot).


	Summary of change
	1) Corrected the updating of the v_TFT2 as shown in the screenshot. Effectively, corrected the position of TOS_TC.
2) Declared a new variable v_TFT1Remove, assigned the correct parameters to this and then sent this as an argument in the Modify PDP Context REQ for Steps 27-28. Please see screenshot. Created a new function f_UTRAN34_ModifyPDP_ContextMT_Accept_11_2_1a, which uses the newly created template cs_ModifyPDP_ContextReq_11_2_1 which includes the PCO that sets the BCM correctly. 
3) Removed the commented code.
4) Also implemented Change 1.1 of TTCN CR R5s150242

	Source of change
	UTRAN34_NISPC.ttcn

	
	


Before:

	function fl_TC_11_2_1a_Body() runs on UTRAN_PTC
  {
    var PDPContextParameters v_PrimPDPContextParameters;
    var PDPContextParameters v_SecPDPContextParameters;
    var IP_VersionType v_IP_VersionType;
    var template (value) QualityOfService v_QualityOfService:= f_UTRAN_InitialiseQoS(utran34_Cell1, cell_DCH_64kPS_RAB_SRB);
    var template (value) TrafficFlowTemplate v_TFT1;
    var template (value) TrafficFlowTemplate v_TFT2;    
    var template (value) TrafficFlowTemplate.packetFilterList v_PacketFilterList;
    var template (value) PacketFilter v_PacketFilterByRef;
    var integer v_PacketFilterLengthByref;
    var integer v_LengthCount := 0;
    var boolean v_IpTypeIsIPv4 := true;
    var octetstring v_IP_Packet;
    //@siclog "Step 1-7" siclog@
    v_PrimPDPContextParameters := f_UTRAN34_ActivateFirstPDP_ContextNISPC(utran34_Cell1, v_QualityOfService, true, true); //check if 2 loops in the TC!
    v_IP_VersionType := v_PrimPDPContextParameters.ip_VersionType;
    if(v_IP_VersionType == iPv6){
        v_IpTypeIsIPv4 := false
    }
    //Generate traffic flow template #1
    f_Get_PacketFilter_10_9_1(v_PacketFilterByRef, v_PacketFilterLengthByref, v_IpTypeIsIPv4, 11 /*R5s150242 Change 1.1 replaced 1->11*/);
    v_PacketFilterList[0]  := v_PacketFilterByRef;
    v_LengthCount := v_LengthCount + v_PacketFilterLengthByref+1;
    v_TFT1 := cs_TrafficFlowTemplate(v_LengthCount, v_PacketFilterList);
    v_TFT1.iei := '36'O;
    //@siclog "Step 8-12" siclog@
    v_SecPDPContextParameters := f_UTRAN34_SecPDP_Context_MT_r7(utran34_Cell1, v_TFT1, v_PrimPDPContextParameters, v_QualityOfService, "Test Case 11.2.1a, Step 9");
    //@siclog "Step 13" siclog@
    //Wait for T3380 seconds to ensure no further activaate request message comes from UE
    f_UTRAN_CheckNoDataInd(tsc_T3380max, "Test Case 11.1.5.2, Step 13");
    //@siclog "Step 14" siclog@
    f_UTRAN34_CloseUE_TestLoop(ps_domain, tsc_UE_TestLoopMode4, cs_UE_TestLoopMode1_LB_Setup2(312, tsc_RB20, 312, tsc_RB22));
    // Route data to UTRAN PTC
    f_IP_Handling_Stop(IP);
    //@siclog "Step 15-16" siclog@
    // IP Packet#1 should NOT match Packet Filter #1
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 2, tsc_RB20, tsc_RB20);
    //@siclog "Step 17-18" siclog@
    // IP Packet#2 should match Packet Filter #1
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 1, tsc_RB20, tsc_RB22);
    //Generate traffic flow template #2
    /*v_LengthCount := 0;
      f_Get_PacketFilter_10_9_1(v_PacketFilterByRef, v_PacketFilterLengthByref, v_IpTypeIsIPv4, 3);
      v_PacketFilterList[0]  := v_PacketFilterByRef;
    v_LengthCount := v_LengthCount + v_PacketFilterLengthByref+1;
    v_TFT2 := cs_TrafficFlowTemplate(v_LengthCount, v_PacketFilterList);
    v_TFT2.iei := '36'O;*/
    v_TFT2 := v_TFT1;
    v_TFT2.tftOperationCode := tsc_TFT_OperationAdd;    
    v_TFT2.packetFilterList[0].contents[0] := cs_PktFilterCompTOSTrafficClass('A0FC'O);

    v_TFT2.packetFilterList[0].precedence := int2bit(5, 8);
    //@siclog "Step 19-20" siclog@
    //SS sends a modify request to UE for secondary PDP context. UE accepts the modification request from SS
    f_UTRAN34_ModifyPDP_ContextMT_Accept(v_SecPDPContextParameters, v_TFT2);
    //@siclog "Step 21-22" siclog@
    // IP Packet#3 (identical to IP Packet#1) should NOT match Packet Filter #1 nor Packet Filter #2
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 2,tsc_RB20, tsc_RB20);
    //@siclog "Step 23-24" siclog@
    // IP Packet#4 (identical to IP Packet#2) should match Packet Filter #1 but not Packet Filter #2
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 1, tsc_RB20, tsc_RB22);
    //@siclog "Step 25-26" siclog@
    // IP Packet#5 should match Packet Filter #2 but not Packet Filter #1
    v_IP_Packet := fl_Get_IP_Packet(v_IpTypeIsIPv4, 'A0'O);
    f_UTRAN34_SendReceiveData(tsc_RB20, tsc_RB22, oct2bit(v_IP_Packet));
    //@siclog "Step 27-28" siclog@
    //Modify Traffic Flow Template #1
    v_TFT1.tftOperationCode := tsc_TFT_OperationRemove;

    //SS sends a modify request to UE for secondary PDP context. UE accepts the modification request from SS
    f_UTRAN34_ModifyPDP_ContextMT_Accept(v_SecPDPContextParameters, v_TFT1);
    //@siclog "Step 29-30" siclog@
    // IP Packet#6 (identical to IP Packet#2) should match Packet Filter #1 but not Packet Filter #2
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 1, tsc_RB20, tsc_RB20);
    //@siclog "Step 31-32" siclog@
    // IP Packet#7 (identical to IP Packet#5) should match Packet Filter #2
    f_UTRAN34_SendReceiveData(tsc_RB20, tsc_RB22, oct2bit(v_IP_Packet));
  }


After:

	function fl_TC_11_2_1a_Body() runs on UTRAN_PTC
  {
    var PDPContextParameters v_PrimPDPContextParameters;
    var PDPContextParameters v_SecPDPContextParameters;
    var IP_VersionType v_IP_VersionType;
    var template (value) QualityOfService v_QualityOfService:= f_UTRAN_InitialiseQoS(utran34_Cell1, cell_DCH_64kPS_RAB_SRB);
    var template (value) TrafficFlowTemplate v_TFT1;
    var template (value) TrafficFlowTemplate v_TFT2;
    var template (value) TrafficFlowTemplate v_TFT1Remove;//WA#11_2_1a
    var template (value) TrafficFlowTemplate.packetFilterList v_PacketFilterList;
    var template (value) PacketFilter v_PacketFilterByRef;
    var integer v_PacketFilterLengthByref;
    var integer v_LengthCount := 0;
    var boolean v_IpTypeIsIPv4 := true;
    var octetstring v_IP_Packet;
    //@siclog "Step 1-7" siclog@
    v_PrimPDPContextParameters := f_UTRAN34_ActivateFirstPDP_ContextNISPC(utran34_Cell1, v_QualityOfService, true, true); //check if 2 loops in the TC!
    v_IP_VersionType := v_PrimPDPContextParameters.ip_VersionType;
    if(v_IP_VersionType == iPv6){
        v_IpTypeIsIPv4 := false
    }
    //Generate traffic flow template #1
    f_Get_PacketFilter_10_9_1(v_PacketFilterByRef, v_PacketFilterLengthByref, v_IpTypeIsIPv4, 11 /*R5s150242 Change 1.1 replaced 1->11*/);
    v_PacketFilterList[0]  := v_PacketFilterByRef;
    v_LengthCount := v_LengthCount + v_PacketFilterLengthByref+1;
    v_TFT1 := cs_TrafficFlowTemplate(v_LengthCount, v_PacketFilterList);
    v_TFT1.iei := '36'O;
    //@siclog "Step 8-12" siclog@
    v_SecPDPContextParameters := f_UTRAN34_SecPDP_Context_MT_r7(utran34_Cell1, v_TFT1, v_PrimPDPContextParameters, v_QualityOfService, "Test Case 11.2.1a, Step 9");
    //@siclog "Step 13" siclog@
    //Wait for T3380 seconds to ensure no further activaate request message comes from UE
    f_UTRAN_CheckNoDataInd(tsc_T3380max, "Test Case 11.2.1a, Step 13");
    //@siclog "Step 14" siclog@
    f_UTRAN34_CloseUE_TestLoop(ps_domain, tsc_UE_TestLoopMode4, cs_UE_TestLoopMode1_LB_Setup2(312, tsc_RB20, 312, tsc_RB22));
    // Route data to UTRAN PTC
    f_IP_Handling_Stop(IP);
    //@siclog "Step 15-16" siclog@
    // IP Packet#1 should NOT match Packet Filter #1
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 2, tsc_RB20, tsc_RB20);
    //@siclog "Step 17-18" siclog@
    // IP Packet#2 should match Packet Filter #1
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 1, tsc_RB20, tsc_RB22);
    //Generate traffic flow template #2
    v_TFT2 := v_TFT1;
    v_TFT2.tftOperationCode := tsc_TFT_OperationAdd;
    v_TFT2.packetFilterList[0].contents[4] := cs_PktFilterCompTOSTrafficClass('A0FC'O);
    // WA#11_2_1a Corrected position of TOS_TC in packet filter above from 0->4
    //v_TFT2.packetFilterList[0].contents[0] := cs_PktFilterCompTOSTrafficClass('A0FC'O);
    v_TFT2.packetFilterList[0].precedence := int2bit(5, 8);
    v_TFT2.packetFilterList[0].iei := '2'H;//WA#11_2_1a
    //@siclog "Step 19-20" siclog@
    //SS sends a modify request to UE for secondary PDP context. UE accepts the modification request from SS
    //WA#11_2_1a f_UTRAN34_ModifyPDP_ContextMT_Accept(v_SecPDPContextParameters, v_TFT2);
    f_UTRAN34_ModifyPDP_ContextMT_Accept_11_2_1a (v_SecPDPContextParameters, v_TFT2);//WA#11_2_1a
    //@siclog "Step 21-22" siclog@
    // IP Packet#3 (identical to IP Packet#1) should NOT match Packet Filter #1 nor Packet Filter #2
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 2,tsc_RB20, tsc_RB20);
    //@siclog "Step 23-24" siclog@
    // IP Packet#4 (identical to IP Packet#2) should match Packet Filter #1 but not Packet Filter #2
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 1, tsc_RB20, tsc_RB22);
    //@siclog "Step 25-26" siclog@
    // IP Packet#5 should match Packet Filter #2 but not Packet Filter #1
    v_IP_Packet := fl_Get_IP_Packet(v_IpTypeIsIPv4, 'A0'O);
    f_UTRAN34_SendReceiveData(tsc_RB20, tsc_RB22, oct2bit(v_IP_Packet));
    //@siclog "Step 27-28" siclog@
    //Modify Traffic Flow Template #1
    //WA#11_2_1a : Generate TFT to Remove Packet Filter #1
    v_TFT1Remove.iei := '36'O;//WA#11_2_1a
    v_TFT1Remove.iel := int2oct(2,1);  //WA#11_2_1a IE Length is 2 octets
    //v_TFT1.tftOperationCode := tsc_TFT_OperationRemove; //WA#11_2_1a
    v_TFT1Remove.tftOperationCode := tsc_TFT_OperationRemove; //WA#11_2_1a
    v_TFT1Remove.eBit := '0'B; //WA#11_2_1a
    v_TFT1Remove.noOfPktFilter := int2bit(1, 4);  //WA#11_2_1a No of Packet Filters is 1
    v_TFT1Remove.packetFilterList[0].spare := '00'B; //WA#11_2_1a
    v_TFT1Remove.packetFilterList[0].direction := '00'B;  //WA#11_2_1a Direction = Spare = 00 for PF Remove
    v_TFT1Remove.packetFilterList[0].iei := '1'H;
    v_TFT1Remove.parameterList := omit; //WA#11_2_1a
    //SS sends a modify request to UE for secondary PDP context. UE accepts the modification request from SS
    //f_UTRAN34_ModifyPDP_ContextMT_Accept(v_SecPDPContextParameters, v_TFT1Remove /*WA#11_2_1a v_TFT1*/);
    f_UTRAN34_ModifyPDP_ContextMT_Accept_11_2_1a (v_SecPDPContextParameters, v_TFT1Remove);//WA#11_2_1a    
    //@siclog "Step 29-30" siclog@
    // IP Packet#6 (identical to IP Packet#2) should match Packet Filter #1 but not Packet Filter #2
    f_UTRAN34_SendReceiveIP_PDU(v_IP_VersionType, 1, tsc_RB20, tsc_RB20);
    //@siclog "Step 31-32" siclog@
    // IP Packet#7 (identical to IP Packet#5) should match Packet Filter #2
    f_UTRAN34_SendReceiveData(tsc_RB20, tsc_RB22, oct2bit(v_IP_Packet));
  }


f_UTRAN34_ModifyPDP_ContextMT_Accept_11_2_1a
	//WA#11_2_1a : New function
  function f_UTRAN34_ModifyPDP_ContextMT_Accept_11_2_1a (PDPContextParameters p_SecondaryPDPContextParameters,
                                                template (omit) TrafficFlowTemplate p_TFT) runs on UTRAN_PTC
  {
    var template (value) LLC_SAPI v_LLC_SAPI := f_UTRAN34_LLC_SAPI_TX(); //@sic R5s140387 sic@
    var PDPContextParameters v_SecPDPContextParameters := p_SecondaryPDPContextParameters;
    v_SecPDPContextParameters.qOS.precedenceClass := '011'B; // Acc to 34.123-3 cl 8.10 @sic R5s140558 sic@
    //Send Modify PDP Context Request message
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_ModPDPContReq(cs_ModifyPDP_ContextReq_11_2_1a (v_SecPDPContextParameters.send_TI,
                                                                        v_LLC_SAPI,
                                                                        v_SecPDPContextParameters.qOS, // @sic R5s140558 sic@
                                                                        p_TFT))));
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U_ModPDPContAcpt(cr_ModifyPDP_ContextAcpt(v_SecPDPContextParameters.received_TI))));
  }


cs_ModifyPDP_ContextReq_11_2_1a 
	template (value) MODIFYPDPCONTEXTREQUEST_dl cs_ModifyPDP_ContextReq_11_2_1a(template (value) TI                  p_TI,
                                                                      template (value) LLC_SAPI            p_LLC_SAPI,
                                                                      template (value) QualityOfService    p_QualityOfService,
                                                                      template (omit)  TrafficFlowTemplate p_TFT) :=
  { /* @descr Modify PDP Context for an UE supporting UMTS only n->ze; TS 24.008, cl 9.5.9 */
    /* @status    APPROVED (UTRAN) */
    ti  := p_TI,
    sM_ProtocolDiscriminator := tsc_SMPD,
    msgType := '01001000'B,
    spare := '0000'B,
    radioPriority := cs_RadioPriorityHigh (omit), //@sic R5s140387 sic@
    requestedLLC_SAPI := p_LLC_SAPI,
    newQoS := p_QualityOfService,
    pDP_Addr := omit,
    pKT_FlowIdentifier := omit,
    protocolConfOpts := cs_ProtocolConfigOptions_OneOption(cs_ProtocolContainer('0005'O, '02'O)),
    tft := p_TFT
  };


5 Branches executed in test case 11.2.1a
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 MediaTek K2 Handset
The MediaTek K2 Handset passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_11_2_1a_MediaTek_K2.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file

